Dubin 4.2.2 Temperature Oscillations 11-15-16

N. T. Gladd

Initialization: Be sure the files *NTGStylesheet2.nb* and *NTGUtilityFunctions.m* is are in the same directory as that from which this notebook was loaded. Then execute the cell immediately below by mousing left on the cell bar to the right of that cell and then typing “shift” + “enter”. Respond “Yes” in response to the query to evaluate initialization cells.

In[1]:=
```
SetDirectory[NotebookDirectory[]];
(* set directory where source files are located *)
SetOptions[EvaluationNotebook[], (* load the StyleSheet *)
 StyleDefinitions → Get["NTGStylesheet2.nb"]];
Get["NTGUtilityFunctions.m"]; (* Load utilities package *)
```

Purpose

I solve a heat equation example from Chapter 4.2.2 of *Numerical and Analytical Methods for Scientists and Engineers, Using Mathematica*, Daniel Dubin. This example treats a time-oscillatory boundary condition.

Solution

I construct an Association that encapsulates information about this problem, and then apply the function *DSolveHeatEquation* that attempts to solve this problem directly using the Mathematica function *DSolve*.

In[6]:=

```
A1 =
  Module[{description, pde, bcL, bcR, ic, eqns, depVar,
    assumptions, substitutions, simplifications, names, values},
   description = "Dubin 4.2.2 Temperature oscillations\nHomogeneous heat
      equation, inhomogeneous time-dependentDirichlet boundary condition";
   pde = D[T[x, t], t] - χ D[T[x, t], {x, 2}] == 0;
   bcL = T[0, t] == T0; (* inhomogeneous Dirichlet *)
   bcR = T[0, t] == T0 + TA Sin[ω t]; (* insulated, homogeneous von Neumann *)
   ic = T[x, 0] == T0;
   eqns = {pde, bcL, bcR, ic};
   depVar = T[x, t];
   assumptions = {L > 0, χ > 0};
   substitutions = {K[1] → n};
   simplifications = {n ∈ Integers};
   values = {description, pde, bcL, bcR, ic,
     eqns, depVar, assumptions, substitutions, simplifications};
   names = {"description", "pde", "bcL", "bcR", "ic", "eqns",
     "depVar", "assumptions", "substitutions", "simplifications"};
   AssociationThread[names, values]];

Module[{soln, G},
 soln = DSolveHeatEquation[A1];
 AppendTo[A1, "soln" → soln];
 Print@ShowPDESetup[A1];
 A1["soln"]]
```

Dubin 4.2.2 Temperature oscillations

Homogeneous heat equation, inhomogeneous time–dependentDirichlet boundary condition



Out[7]= $T^{(0,1)}[x,t] == \chi\, T^{(2,0)}[x,t]$

This problem is not immediately handled by DSolve.

The solution strategy is to write T(x,t) as the sum of a term that explicitly satisfies the boundary conditions T_{bc}(x) and a term that has homogeneous boundary conditions T_h(x,t).

In[8]:=

```
w[1] = T[x, t] == Tbc[x, t] + Th[x, t]
```

Out[8]= $T[x,t] == T_{bc}[x,t] + T_h[x,t]$

A suitable choice for T_{bc} is

In[9]:=
```
w[2] = Tbc[x, t] == T0 (1 - x/L) + x/L (T0 + TA Sin[ω t]) // Expand
```
Out[9]=
$$T_{bc}[x, t] == T_0 + \frac{x \operatorname{Sin}[t\,\omega]\, T_A}{L}$$

The transformed heat equation is

In[10]:=
```
w[3] = A1["pde"] /. T → Function[{x, t}, Th[x, t] + T0 + (x Sin[t ω] TA)/L];
w[3] = MapEqn[(# - (x ω Cos[t ω] TA)/L) &, w[3]]
```
Out[11]=
$$T_h{}^{(0,1)}[x, t] - \chi\, T_h{}^{(2,0)}[x, t] == -\frac{x\,\omega \operatorname{Cos}[t\,\omega]\, T_A}{L}$$

The boundary conditions for T_h(x, t) are

In[12]:=
```
w[4] = {Th[0, t] == 0, Th[L, t] == 0}
```
Out[12]=
$$\{T_h[0, t] == 0, T_h[L, t] == 0\}$$

The new initial condition for T_h(x, t) is

In[13]:=
```
w[4] =
 Th[x, t] == T[x, t] - Tbc[x, t] /. t → 0 /. T[x, 0] → T0 /. (w[2] /. t → 0 // ER)
```
Out[13]=
$$T_h[x, 0] == 0$$

A solution technique for such problems is to represent the source term and initial condition in terms of expansions of the eigenfunctions of
the homogeneous equation for T_h(x,t). Separation of variables leads to

In[14]:=
```
w[5] = (A1["pde"][[1]] /. T → Th) == 0 /. Th → Function[{x, t}, 𝒯[t] ψ[x]];
w[5] = MapEqn[(# / (𝒯[t] ψ[x])) &, w[5]] // Expand
```
Out[15]=
$$\frac{\mathcal{T}'[t]}{\mathcal{T}[t]} - \frac{\chi\,\psi''[x]}{\psi[x]} == 0$$

The separated equations are

In[16]:=
```
w[6] = {w[5][[1, 1]] == -λ, w[5][[1, 2]] == λ}
```
Out[16]=
$$\left\{\frac{\mathcal{T}'[t]}{\mathcal{T}[t]} == -\lambda,\ -\frac{\chi\,\psi''[x]}{\psi[x]} == \lambda\right\}$$

The latter constitutes a Sturm-Liouville equation (see Appendix A)

In[17]:=
```
w[7] = {λn → n^2 π^2 χ/L^2, ψn[x] → Sin[x Sqrt[λn]/Sqrt[χ]]}
```

Out[17]= $\left\{\lambda_n \to \frac{n^2 \pi^2 \chi}{L^2},\ \psi_n[x] \to \text{Sin}\left[\frac{x\sqrt{\lambda_n}}{\sqrt{\chi}}\right]\right\}$

Then, the solution for the homogeneous equation for T_h(x,t) is

In[18]:=
```
w[8] = Th[x, t] == Sum[𝒯n[t] ψn[x], {n, 1, ∞}]
```

Out[18]= $T_h[x, t] == \sum_{n=1}^{\infty} \mathcal{T}_n[t]\, \psi_n[x]$

Substitute this into the inhomogeneous equation

In[19]:=
```
w[9] = w[3] /. Th → Function[{x, t}, Sum[𝒯n[t] ψn[x], {n, 1, ∞}]]
```

Out[19]= $\sum_{n=1}^{\infty} \psi_n[x]\, \mathcal{T}_n'[t] - \chi \sum_{n=1}^{\infty} \mathcal{T}_n[t]\, \psi_n''[x] == -\frac{x\, \omega\, \text{Cos}[t\, \omega]\, T_A}{L}$

In[20]:=
```
w[10] = w[9] /. ψn''[x] → -λn/χ ψn[x]
```

Out[20]= $-\chi \sum_{n=1}^{\infty} -\frac{\lambda_n \mathcal{T}_n[t]\, \psi_n[x]}{\chi} + \sum_{n=1}^{\infty} \psi_n[x]\, \mathcal{T}_n'[t] == -\frac{x\, \omega\, \text{Cos}[t\, \omega]\, T_A}{L}$

Write the source term in terms of an eigenfunction expansion

In[21]:=
```
w[11] = w[10] /. -x ω Cos[t ω] TA/L → Sum[fn[t] ψn[x], {n, 1, ∞}]
```

Out[21]= $-\chi \sum_{n=1}^{\infty} -\frac{\lambda_n \mathcal{T}_n[t]\, \psi_n[x]}{\chi} + \sum_{n=1}^{\infty} \psi_n[x]\, \mathcal{T}_n'[t] == \sum_{n=1}^{\infty} f_n[t]\, \psi_n[x]$

For each n

In[22]:=
```
w[12] = w[11] /. Sum[a_, b_] → a;
w[12] = MapEqn[(# / ψn[x]) &, w[12]] // Expand
```

Out[23]= $\lambda_n \mathcal{T}_n[t] + \mathcal{T}_n'[t] == f_n[t]$

The explicit f_n[t] are given by

In[24]:=
$$w[13] = f_n[t] == \frac{\int_0^L -\frac{x\,\omega\,\mathrm{Cos}[t\,\omega]\,T_A}{L}\,\mathrm{Sin}\left[\frac{n\,\pi\,x}{L}\right] dx}{\int_0^L \mathrm{Sin}\left[\frac{n\,\pi\,x}{L}\right]^2 dx} \text{ // Refine[\#, } n \in \text{Integers] \&}$$

Out[24]=
$$f_n[t] == \frac{2\,(-1)^n\,\omega\,\mathrm{Cos}[t\,\omega]\,T_A}{n\,\pi}$$

Thus

In[25]:=
$$w[14] = w[12] \text{ /. } (w[13] \text{ // ER})$$

Out[25]=
$$\lambda_n\,\mathcal{T}_n[t] + \mathcal{T}_n'[t] == \frac{2\,(-1)^n\,\omega\,\mathrm{Cos}[t\,\omega]\,T_A}{n\,\pi}$$

In[26]:=
$$w[15] = \mathrm{DSolve}[w[14], \mathcal{T}_n[t], t][[1, 1]] \text{ /. } C[1] \to A_n$$

Out[26]=
$$\mathcal{T}_n[t] \to e^{-t\,\lambda_n}\,A_n + \left(2\,(-1)^n\,\omega\,T_A\,(\omega\,\mathrm{Sin}[t\,\omega] + \mathrm{Cos}[t\,\omega]\,\lambda_n)\right) / \left(n\,\pi\,(\omega^2 + \lambda_n^2)\right)$$

The value of the constant A_n are determined by the initial condition

In[27]:=
$$w[16] = T_h[x, t] == \sum_{n=1}^{\infty} \mathcal{T}_n[t]\,\psi_n[x] \text{ /. } w[15]$$

Out[27]=
$$T_h[x, t] == \sum_{n=1}^{\infty} \left(e^{-t\,\lambda_n}\,A_n + \left(2\,(-1)^n\,\omega\,T_A\,(\omega\,\mathrm{Sin}[t\,\omega] + \mathrm{Cos}[t\,\omega]\,\lambda_n)\right) / \left(n\,\pi\,(\omega^2 + \lambda_n^2)\right)\right)\psi_n[x]$$

In[28]:=
$$w[17] = w[16] \text{ /. } t \to 0 \text{ /. } T_h[x, 0] \to 0$$

Out[28]=
$$0 == \sum_{n=1}^{\infty} \left(A_n + \frac{2\,(-1)^n\,\omega\,T_A\,\lambda_n}{n\,\pi\,(\omega^2 + \lambda_n^2)}\right)\psi_n[x]$$

Thus, each term of the summand must be zero, and the explicit A_n are

In[29]:=
$$w[18] = \mathrm{Solve}\left[A_n + \frac{2\,(-1)^n\,\omega\,T_A\,\lambda_n}{n\,\pi\,(\omega^2 + \lambda_n^2)} == 0, A_n\right][[1, 1]]$$

Out[29]=
$$A_n \to -\frac{2\,(-1)^n\,\omega\,T_A\,\lambda_n}{n\,\pi\,(\omega^2 + \lambda_n^2)}$$

So, the complete solution for the temporal term is

In[30]:=
$$w[19] = w[15] \text{ /. } w[18] \text{ // Simplify}$$

Out[30]=
$$\mathcal{T}_n[t] \to \left(2\,(-1)^n\,e^{-t\,\lambda_n}\,\omega\,T_A\,\left(e^{t\,\lambda_n}\,\omega\,\mathrm{Sin}[t\,\omega] + (-1 + e^{t\,\lambda_n}\,\mathrm{Cos}[t\,\omega])\,\lambda_n\right)\right) / \left(n\,\pi\,(\omega^2 + \lambda_n^2)\right)$$

and

In[31]:=
```
w[20] = w[8] /. Sum → Inactive[Sum] /. w[19] //. w[7]
```

Out[31]=
$$T_h[x,t] == \sum_{n=1}^{\infty} \frac{2\,(-1)^n\, e^{-\frac{n^2 \pi^2 t \chi}{L^2}}\, \omega\, \mathrm{Sin}\left[\frac{\pi\, x \sqrt{\frac{n^2 \chi}{L^2}}}{\sqrt{\chi}}\right] \left(\frac{n^2 \pi^2 \chi \left(-1 + e^{\frac{n^2 \pi^2 t \chi}{L^2}} \mathrm{Cos}[t\,\omega]\right)}{L^2} + e^{\frac{n^2 \pi^2 t \chi}{L^2}}\, \omega\, \mathrm{Sin}[t\,\omega]\right) T_A}{n\,\pi \left(\frac{n^4 \pi^4 \chi^2}{L^4} + \omega^2\right)}$$

Finally, the solution of the original equation is

In[32]:=
```
w[21] = w[1] /. (w[2] // ER) /. (w[20] // ER)
```

Out[32]=
$$T[x,t] == T_0 + \frac{x\, \mathrm{Sin}[t\,\omega]\, T_A}{L} + \sum_{n=1}^{\infty} \frac{2\,(-1)^n\, e^{-\frac{n^2 \pi^2 t \chi}{L^2}}\, \omega\, \mathrm{Sin}\left[\frac{\pi\, x \sqrt{\frac{n^2 \chi}{L^2}}}{\sqrt{\chi}}\right] \left(\frac{n^2 \pi^2 \chi \left(-1 + e^{\frac{n^2 \pi^2 t \chi}{L^2}} \mathrm{Cos}[t\,\omega]\right)}{L^2} + e^{\frac{n^2 \pi^2 t \chi}{L^2}}\, \omega\, \mathrm{Sin}[t\,\omega]\right) T_A}{n\,\pi \left(\frac{n^4 \pi^4 \chi^2}{L^4} + \omega^2\right)}$$

I eliminate subscripts in preparation for a numerical treatment

In[33]:=
```
w[21] /. T₀ → T0 /. T_A → TA /. ∞ → nMax
```

Out[33]=
$$T[x,t] == T0 + \frac{TA\, x\, \mathrm{Sin}[t\,\omega]}{L} + \sum_{n=1}^{nMax} \frac{2\,(-1)^n\, e^{-\frac{n^2 \pi^2 t \chi}{L^2}}\, TA\, \omega\, \mathrm{Sin}\left[\frac{\pi\, x \sqrt{\frac{n^2 \chi}{L^2}}}{\sqrt{\chi}}\right] \left(\frac{n^2 \pi^2 \chi \left(-1 + e^{\frac{n^2 \pi^2 t \chi}{L^2}} \mathrm{Cos}[t\,\omega]\right)}{L^2} + e^{\frac{n^2 \pi^2 t \chi}{L^2}}\, \omega\, \mathrm{Sin}[t\,\omega]\right)}{n\,\pi \left(\frac{n^4 \pi^4 \chi^2}{L^4} + \omega^2\right)}$$

In[34]:=
```
Clear[TSoln];
TSoln[x_, t_, L_, χ_, T0_, TA_, ω_, nMax_] :=
```
$$T0 + \frac{TA\, x\, \mathrm{Sin}[t\,\omega]}{L} + \sum_{n=1}^{nMax} \frac{2\,(-1)^n\, e^{-\frac{n^2 \pi^2 t \chi}{L^2}}\, TA\, \omega\, \mathrm{Sin}\left[\frac{\pi\, x \sqrt{\frac{n^2 \chi}{L^2}}}{\sqrt{\chi}}\right] \left(\frac{n^2 \pi^2 \chi \left(-1 + e^{\frac{n^2 \pi^2 t \chi}{L^2}} \mathrm{Cos}[t\,\omega]\right)}{L^2} + e^{\frac{n^2 \pi^2 t \chi}{L^2}}\, \omega\, \mathrm{Sin}[t\,\omega]\right)}{n\,\pi \left(\frac{n^4 \pi^4 \chi^2}{L^4} + \omega^2\right)} \;//$$
```
  Activate
```

For the parameters specifies in Dubin's example (p287)

In[36]:=
```
Module[{L = 4, χ = 1/8, T0 = 2, TA = 1, ω = 1, nMax = 5},
 Plot3D[TSoln[x, t, L, χ, T0, TA, ω, nMax], {x, 0, 4}, {t, 0, 10}, PlotRange → All,
  ColorFunction → "TemperatureMap", AxesLabel → {Stl["x"], Stl["t"], Stl["T"]},
  Mesh → False, Boxed → False, PlotLegends → Automatic]]
```

Out[36]=



I check this result by solving the same problem numerically

In[37]:=
```
SOLNNUMERICAL = Module[{L = 4, χ = 1/8, T0 = 2, TA = 1, ω = 1},
   NDSolve[{D[T[x, t], t] == χ D[T[x, t], {x, 2}], T[0, t] == T0,
     T[L, t] == T0 + TA Sin[ω t], T[x, 0] == T0}, T, {x, 0, L}, {t, 0, 10}]][[1, 1]]
```

Out[37]=
```
T → InterpolatingFunction[ Domain: {{0., 4.}, {0., 10.}}
                           Output: scalar ]
```

I compare TSoln against the numerical solutions as a function of time at the point x = 3.75.

In[38]:=
```
Module[{L = 4, χ = 1/8, T0 = 2, TA = 1, ω = 1, nMax = 10, xTest = 3.75},
 Plot[
  {T[xTest, t] /. SOLNNUMERICAL, TSoln[xTest, t, L, χ, T0, TA, ω, nMax]}, {t, 0, 10},
  PlotStyle → {Gray, Directive[Red, Dashed]},
  PlotLegends → {"Numerical", "Analytic"}, AxesLabel →
   {Stl["t"], Stl["T[x = 3.75,t]"]}, PlotLabel → Stl["Agreement is excellent"]]]
```

Out[38]=



Appendix Spatial eigenvalue problem

In[39]:= wA[1] = DSolve[$-\frac{\chi\,\psi''[x]}{\psi[x]}$ == λ, ψ[x], x][[1, 1]] // RE

Out[39]= $\psi[x] == C[1]\,\text{Cos}\left[\frac{x\sqrt{\lambda}}{\sqrt{\chi}}\right] + C[2]\,\text{Sin}\left[\frac{x\sqrt{\lambda}}{\sqrt{\chi}}\right]$

In[40]:=
```
wA[2] = wA[1] /. x → 0 /. ψ[0] → 0
```

Out[40]= 0 == C[1]

In[41]:=
```
wA[3] = wA[1] /. Solve[wA[2], C[1]][[1]]
```

Out[41]= $\psi[x] == C[2]\,\text{Sin}\left[\frac{x\sqrt{\lambda}}{\sqrt{\chi}}\right]$

The eigenvalues are given by

In[42]:=
```
wA[4] = Solve[(x Sqrt[λ])/Sqrt[χ] == n π, λ][[1, 1]] /. λ → λn
```

Out[42]=
$$\lambda_n \to \frac{n^2 \pi^2 \chi}{x^2}$$

Functions

In[3]:=
```
Clear[ShowPDESetup];
ShowPDESetup[A_] := Module[{top = 1.0, right = 1.0,
   boundaries, labels, textInterior, textIC, textBCL, textBCR},
  boundaries = Line /@ {{{0, 0}, {right, 0}},
     {{0, 0}, {0, top}}, {{right, 0}, {right, top}}};
  labels = Text[PhysicsForm[A[#[[1]]] ], #[[2]]] & /@
    {{"pde", {right / 2, top / 2}}, {"ic", {right / 2, 0.0}},
     {"bcL", {0.0, top / 2}}, {"bcR", {right, top / 2}}};
  Graphics[{Directive[Black, Thick], boundaries, labels}, Axes → False,
   AspectRatio → 0.25, ImageSize → 500, PlotLabel → Stl[A["description"]]]]
```

In[4]:=
```
Clear[DSolveHeatEquation];
DSolveHeatEquation[A_] :=
 Module[{soln},
  soln =
   DSolve[A["eqns"], A["depVar"], {x, t}, Assumptions → A["assumptions"]][[1, 1]];
  soln = soln //. A["substitutions"];
  soln = Simplify[soln, A["simplifications"]];
  soln]
```